

Comment: Copyright (c) 2010, 2016 and 2017 by Murder Accountability Project.

Comment: These SPSS commands open the current Murder Accountability Project (MAP) enhanced Supplemental Homicide Report file and calculate if police knew who committed each of the more than 732,000 homicides by examining the reported gender of the offender; FBI statisticians use "U" to indicate the offender's gender was unknown.

```
GET FILE = '/PATH/SHR1976_2015.SAV'.  
EXECUTE.  
COMPUTE SOLVED=1.  
IF (OFFSEX = "U") SOLVED=0.  
VALUE LABELS SOLVED 0 "Not Solved" 1 "Solved".  
EXECUTE.
```

Comment: This creates a numeric coding for the victim's gender.

```
COMPUTE SEX =9.  
IF (VICSEX = "M") SEX=1.  
IF (VICSEX = "F") SEX=2.  
VALUE LABELS SEX 1 "Male" 2 "Female" 9 "Unknown".  
EXECUTE.
```

Comment: These commands create two clustering variables, MURDGRP1, an 8-digit number based upon each homicide's county location, victim sex and method of killing and MURDGRP2, a 7-digit number based upon each homicide's metropolitan area, victim sex and method of killing.

```
COMPUTE CNTY=NUMBER (CNTYFIPS, F5.0).  
COMPUTE MURDGRP1 = (CNTY * 1000) + (SEX * 100) + WEAPON.  
EXECUTE.  
FORMAT MURDGRP1 (F8.0).  
COMPUTE MURDGRP2 = (MSA * 1000) + (SEX * 100) + WEAPON.  
EXECUTE.  
FORMAT MURDGRP2 (F7.0).
```

Comment: This aggregates the number of homicides, number of solved homicides and percentage of solved homicides by MURDGRP and creates a new file of aggregated data called MURDERGROUP1 and MURDERGROUP2.

```
AGGREGATE  
  /OUTFILE='/PATH/MURDERGROUP1.sav'  
  /BREAK=MURDGRP1 SEX CNTY WEAPON  
  /TOTAL = NU(SOLVED) /SOLVED = SUM(SOLVED) /PERCENT = MEAN(SOLVED).  
EXECUTE.  
AGGREGATE  
  /OUTFILE='/PATH/MURDERGROUP2.sav'
```

```
/BREAK=MURDGRP2 SEX MSA WEAPON  
/TOTAL = NU(SOLVED) /SOLVED = SUM(SOLVED) /PERCENT = MEAN(SOLVED).  
EXECUTE.
```

Comment: This calls up the MURDERGROUP1 file (or MURDERGROUP2 if desired), selects only female victims, eliminates aggregates with incomplete information (such as murders committed outside of an MSA or cases in which key information was not reported), eliminates aggregates in which more than 33 percent of the homicides were solved, computes how many cases were unsolved and sorts by the total number of unsolved cases in descending order, most to least.

```
GET FILE = '/Path/MURDERGROUP1.sav'.  
EXECUTE.  
SELECT IF (SEX =2).  
EXECUTE.  
SELECT IF (MURDGRP> 0).  
EXECUTE.  
SELECT IF (PERCENT <= .33).  
EXECUTE.  
COMPUTE UNSOLVED = TOTAL - SOLVED.  
EXECUTE.  
SORT CASES BY UNSOLVED (D).  
EXECUTE.
```

Comment: After these operations are completed, what will remain are hundreds of clusters of cities that have had unusually large numbers of unsolved homicides involving women; This process can be repeated to focus on male victims as well.